

Algebraic evaluation of error independent classifiers on unlabeled data

The core mathematical theorem of our submission is explained here.

The Polynomial System - Algebra of Error for Independent Binary Classifiers

The following set of quartic polynomials define algebraic system that must be solved to estimate the accuracy of three classifiers as well as the true prevalence of the labels. There are $8 = 2^3$ polynomials, one for each of the possible voting pattern of three classifiers doing binary label classification. They define a polynomial ideal that we call an “evaluation ideal.”

```
In[2197]:= evaluationIdeal = MakeIndependentVotingIdeal[{1, 2, 3}]
Out[2197]= {Pα P1,α P2,α P3,α + (1 - Pα) (1 - P1,β) (1 - P2,β) (1 - P3,β) - fα,α,α,
Pα P1,α P2,α (1 - P3,α) + (1 - Pα) (1 - P1,β) (1 - P2,β) P3,β - fα,α,β,
Pα P1,α (1 - P2,α) P3,α + (1 - Pα) (1 - P1,β) P2,β (1 - P3,β) - fα,β,α,
Pα P1,α (1 - P2,α) (1 - P3,α) + (1 - Pα) (1 - P1,β) P2,β P3,β - fα,β,β,
Pα (1 - P1,α) P2,α P3,α + (1 - Pα) P1,β (1 - P2,β) (1 - P3,β) - fβ,α,α,
Pα (1 - P1,α) P2,α (1 - P3,α) + (1 - Pα) P1,β (1 - P2,β) P3,β - fβ,α,β,
Pα (1 - P1,α) (1 - P2,α) P3,α + (1 - Pα) P1,β P2,β (1 - P3,β) - fβ,β,α,
Pα (1 - P1,α) (1 - P2,α) (1 - P3,α) + (1 - Pα) P1,β P2,β P3,β - fβ,β,β}
```

Evaluation on unlabeled data

In classification problems, the ground truth is the correct label for each item in a dataset. The statistics of ground truth solved by the above polynomial system are:

- the prevalence of the labels.
- the accuracies of the classifiers for each label.

Note that these are sample statistics. The method is not trying to infer anything about the process that created the data over which the binary classifiers have been run. There is some data, you ran binary classifiers over it and now you want to ask statistical questions about their decisions over that data. It

would be trivial to answer them IF you had the true labels for each of the data points in your dataset. But you don't have the ground truth so you need techniques like the one explained here.

Creating error independent binary classifiers

The purpose of this notebook is to demonstrate the correctness of exact algebraic evaluation in the somewhat unrealistic case of error independent binary classifiers. But although unrealistic, this exact solution is important theoretically. For one, exact solutions of anything are rare. A class of problems that frequently admit exact solution are those that assume some sort of independence. We are doing the same here. This unrealistic solution will turn out to be very useful in practice.

The following commands create the ground truth for this problem - a list of labels for the true label of each data point in the dataset.

```
In[2198]:= desiredALabelPrevalence = 23 / 100
datasetSize = 10 000
labelGroundTruth =
  Table[If[RandomReal[] < desiredALabelPrevalence,  $\alpha$ ,  $\beta$ ], {datasetSize}];
Out[2198]=  $\frac{23}{100}$ 
Out[2199]= 10 000
```

Let's check that the ground truth makes sense

```
In[2201]:= RandomSample[labelGroundTruth, 10]
Out[2201]= { $\beta$ ,  $\beta$ ,  $\beta$ ,  $\beta$ ,  $\beta$ ,  $\beta$ ,  $\beta$ ,  $\beta$ ,  $\beta$ ,  $\beta$ }
In[2202]:= labelGroundTruth // Tally
Out[2202]= {{ $\alpha$ , 2253}, { $\beta$ , 7747}}
```

Now we construct the synthetic data for the classifier label decisions. Again, we are using synthetic data to create a simple example that focuses on the math of the algorithm. You can substitute numbers from real independent classifiers if you have them.

1. The first thing we need to specify is the accuracy of each of the classifiers. This accuracy is one of the statistics we seek to estimate with the polynomial system that is constructed from the classifier decisions alone. I'll arbitrarily pick them uniformly from the same range. I need two accuracies, one for each label since we are doing binary classification.

```
In[2203]:= classifierAccuracies = Table[
  {(* The  $\alpha$  accuracy *)  $\alpha \rightarrow$  RandomReal[{0.7, 0.9}],
    (* The  $\beta$  accuracy *)  $\beta \rightarrow$  RandomReal[{0.7, 0.9}]} // Association, {3}]
Out[2203]= {<|  $\alpha \rightarrow 0.797697$ ,  $\beta \rightarrow 0.819408$  |>,
  <|  $\alpha \rightarrow 0.738878$ ,  $\beta \rightarrow 0.888438$  |>, <|  $\alpha \rightarrow 0.717426$ ,  $\beta \rightarrow 0.757177$  |> }
```

2. Now we need a function that takes a classifier's accuracies, a list true labels, and produces a sample of its classification decisions.

- OtherLabel is a convenience function to create incorrect classification decisions.
- SyntheticClassification is the actual function that produces a sample of what a classifier with the specified accuracies would produce on a dataset.

```
In[2204]:= OtherLabel[ $\alpha$ ] =  $\beta$ 
OtherLabel[ $\beta$ ] =  $\alpha$ 
```

```
Out[2204]=  $\beta$ 
```

```
Out[2205]=  $\alpha$ 
```

```
In[2206]:= SyntheticClassification[classifierAccuracies_Association, dataset_List] :=
dataset //
Map[
  If[
    (* Throw a die, if below accuracy,
    pick the true label, if not, pick the other label *)
    RandomReal[] < classifierAccuracies[#, #, OtherLabel[#]] &,
    #] &
```

3. The classifier decisions are next. Three lists, one for each classifier, as long as the dataset.

```
In[2207]:= classifierDecisions =
Map[SyntheticClassification[#, labelGroundTruth] &, classifierAccuracies];
```

Let's make sure things are okay

```
In[2208]:= classifierDecisions[[1]] // {RandomSample[#, 10], # // Length} &
classifierDecisions[[2]] // {RandomSample[#, 10], # // Length} &
classifierDecisions[[3]] // {RandomSample[#, 10], # // Length} &
```

```
Out[2208]= {{ $\beta$ ,  $\beta$ ,  $\beta$ ,  $\beta$ ,  $\beta$ ,  $\alpha$ ,  $\alpha$ ,  $\alpha$ ,  $\beta$ ,  $\alpha$ }, 10 000}
```

```
Out[2209]= {{ $\alpha$ ,  $\beta$ ,  $\alpha$ ,  $\alpha$ ,  $\beta$ ,  $\beta$ ,  $\alpha$ ,  $\alpha$ ,  $\beta$ ,  $\beta$ }, 10 000}
```

```
Out[2210]= {{ $\beta$ ,  $\alpha$ ,  $\beta$ ,  $\beta$ ,  $\alpha$ ,  $\beta$ ,  $\alpha$ ,  $\alpha$ ,  $\beta$ ,  $\beta$ }, 10 000}
```

Let's also make sure that we are getting noisy classification of the true labels

```
In[2211]:= Map[Tally, classifierDecisions]
```

```
Out[2211]= {{{ $\alpha$ , 3186}, { $\beta$ , 6814}}, {{ $\alpha$ , 2509}, { $\beta$ , 7491}}, {{ $\alpha$ , 3437}, { $\beta$ , 6563}}}
```

4. Now we can construct the left side input of the ground truth polynomials for independent binary classifiers - the frequency of the voting patterns

```
In[2212]:= votingPatternCounts = classifierDecisions // Transpose // Tally
```

```
Out[2212]= {{{ $\alpha$ ,  $\alpha$ ,  $\alpha$ }, 982}, {{ $\beta$ ,  $\beta$ ,  $\beta$ }, 4355}, {{ $\alpha$ ,  $\alpha$ ,  $\beta$ }, 474}, {{ $\beta$ ,  $\beta$ ,  $\alpha$ }, 1406},
{{ $\alpha$ ,  $\beta$ ,  $\beta$ }, 1107}, {{ $\beta$ ,  $\alpha$ ,  $\beta$ }, 627}, {{ $\beta$ ,  $\alpha$ ,  $\alpha$ }, 426}, {{ $\alpha$ ,  $\beta$ ,  $\alpha$ }, 623}}
```

Let's sort them and pretty print them to see the pattern better

```
In[2213]:= votingPatternCounts // Sort // Column
```

```
{ {α, α, α}, 982 }
```

```
{ {α, α, β}, 474 }
```

```
{ {α, β, α}, 623 }
```

```
{ {α, β, β}, 1107 }
```

```
Out[2213]=
```

```
{ {β, α, α}, 426 }
```

```
{ {β, α, β}, 627 }
```

```
{ {β, β, α}, 1406 }
```

```
{ {β, β, β}, 4355 }
```

Everything looks okay with our synthetic data. Note that `votingPatternCounts` no longer contains any information about the true label for any of the data points in `labelGroundTruth`. Hence, any algorithm that only uses `votingPatternCounts` - the observed frequency of classifier voting patterns, would be carrying out ground truth inference. The goal of the next section is to estimate classifier accuracies and the prevalence of the labels using only `votingPatternCounts`.

Exact Solution for Three Independent Binary Classifiers

Let's assemble all the pieces needed to carry out evaluation. We first take a look at the variables in the evaluation ideal and make sure we have the correct notation.

```
In[2217]:= Variables /@ evaluationIdeal // Sort /@ # & // Column[#, Dividers → All] &
```

```
Out[2217]=
```

$\{P_\alpha, P_{1,\alpha}, P_{1,\beta}, P_{2,\alpha}, P_{2,\beta}, P_{3,\alpha}, P_{3,\beta}, f_{\alpha,\alpha,\alpha}\}$
$\{P_\alpha, P_{1,\alpha}, P_{1,\beta}, P_{2,\alpha}, P_{2,\beta}, P_{3,\alpha}, P_{3,\beta}, f_{\alpha,\alpha,\beta}\}$
$\{P_\alpha, P_{1,\alpha}, P_{1,\beta}, P_{2,\alpha}, P_{2,\beta}, P_{3,\alpha}, P_{3,\beta}, f_{\alpha,\beta,\alpha}\}$
$\{P_\alpha, P_{1,\alpha}, P_{1,\beta}, P_{2,\alpha}, P_{2,\beta}, P_{3,\alpha}, P_{3,\beta}, f_{\alpha,\beta,\beta}\}$
$\{P_\alpha, P_{1,\alpha}, P_{1,\beta}, P_{2,\alpha}, P_{2,\beta}, P_{3,\alpha}, P_{3,\beta}, f_{\beta,\alpha,\alpha}\}$
$\{P_\alpha, P_{1,\alpha}, P_{1,\beta}, P_{2,\alpha}, P_{2,\beta}, P_{3,\alpha}, P_{3,\beta}, f_{\beta,\alpha,\beta}\}$
$\{P_\alpha, P_{1,\alpha}, P_{1,\beta}, P_{2,\alpha}, P_{2,\beta}, P_{3,\alpha}, P_{3,\beta}, f_{\beta,\beta,\alpha}\}$
$\{P_\alpha, P_{1,\alpha}, P_{1,\beta}, P_{2,\alpha}, P_{2,\beta}, P_{3,\alpha}, P_{3,\beta}, f_{\beta,\beta,\beta}\}$

Note the extreme symmetry of these evaluation polynomials. All the members of the generating ideal shown here have exactly the same unknown sample statistics variables.

Preparing the observed voting frequencies rules

Recall that we have the voting pattern counts already

```
In[2218]:= votingPatternCounts // Sort // Column
```

```
{ {α, α, α}, 982 }
{ {α, α, β}, 474 }
{ {α, β, α}, 623 }
{ {α, β, β}, 1107 }
Out[2218]= { {β, α, α}, 426 }
{ {β, α, β}, 627 }
{ {β, β, α}, 1406 }
{ {β, β, β}, 4355 }
```

Let's turn them into rules so we can substitute their value into the ground truth polynomials. We need to normalize them so we need the total count first. We know it is 10K, but let's check it by computation so we can verify everything is okay

```
In[2219]:= totalVotes = votingPatternCounts // Last /@# & // Total
totalVotes == datasetSize
```

```
Out[2219]= 10 000
```

```
Out[2220]= True
```

```
In[2221]:= frequencyRules = votingPatternCounts //
Map[(Subscript[f, Sequence@@#][1]] → #[[2]] / datasetSize) &, #] &
```

```
Out[2221]= { fα,α,α →  $\frac{491}{5000}$ , fβ,β,β →  $\frac{871}{2000}$ , fα,α,β →  $\frac{237}{5000}$ , fβ,β,α →  $\frac{703}{5000}$ ,
fα,β,β →  $\frac{1107}{10000}$ , fβ,α,β →  $\frac{627}{10000}$ , fβ,α,α →  $\frac{213}{5000}$ , fα,β,α →  $\frac{623}{10000}$  }
```

The final polynomial set

```
In[2222]:= finalPolynomialSet = evaluationIdeal /. frequencyRules
```

$$\begin{aligned} \text{Out[2222]} = & \left\{ -\frac{491}{5000} + P_{\alpha} P_{1,\alpha} P_{2,\alpha} P_{3,\alpha} + (1 - P_{\alpha}) (1 - P_{1,\beta}) (1 - P_{2,\beta}) (1 - P_{3,\beta}), \right. \\ & -\frac{237}{5000} + P_{\alpha} P_{1,\alpha} P_{2,\alpha} (1 - P_{3,\alpha}) + (1 - P_{\alpha}) (1 - P_{1,\beta}) (1 - P_{2,\beta}) P_{3,\beta}, \\ & -\frac{623}{10000} + P_{\alpha} P_{1,\alpha} (1 - P_{2,\alpha}) P_{3,\alpha} + (1 - P_{\alpha}) (1 - P_{1,\beta}) P_{2,\beta} (1 - P_{3,\beta}), \\ & -\frac{1107}{10000} + P_{\alpha} P_{1,\alpha} (1 - P_{2,\alpha}) (1 - P_{3,\alpha}) + (1 - P_{\alpha}) (1 - P_{1,\beta}) P_{2,\beta} P_{3,\beta}, \\ & -\frac{213}{5000} + P_{\alpha} (1 - P_{1,\alpha}) P_{2,\alpha} P_{3,\alpha} + (1 - P_{\alpha}) P_{1,\beta} (1 - P_{2,\beta}) (1 - P_{3,\beta}), \\ & -\frac{627}{10000} + P_{\alpha} (1 - P_{1,\alpha}) P_{2,\alpha} (1 - P_{3,\alpha}) + (1 - P_{\alpha}) P_{1,\beta} (1 - P_{2,\beta}) P_{3,\beta}, \\ & -\frac{703}{5000} + P_{\alpha} (1 - P_{1,\alpha}) (1 - P_{2,\alpha}) P_{3,\alpha} + (1 - P_{\alpha}) P_{1,\beta} P_{2,\beta} (1 - P_{3,\beta}), \\ & \left. -\frac{871}{2000} + P_{\alpha} (1 - P_{1,\alpha}) (1 - P_{2,\alpha}) (1 - P_{3,\alpha}) + (1 - P_{\alpha}) P_{1,\beta} P_{2,\beta} P_{3,\beta} \right\} \end{aligned}$$

Let's check that only the seven unknown statistics of the ground truth remain in the final polynomial system.

```
In[2223]:= statisticsOfGroundTruth =
```

```
finalPolynomialSet // Variables /@ # & // Flatten // DeleteDuplicates
```

$$\text{Out[2223]} = \{P_{\alpha}, P_{1,\alpha}, P_{2,\alpha}, P_{3,\alpha}, P_{1,\beta}, P_{2,\beta}, P_{3,\beta}\}$$

The “magical” solution

In[2227]:= `sols = Solve[Map[# == 0 &, finalPolynomialSet], statisticsOfGroundTruth]`

Out[2227]=
$$\left\{ \left\{ P_{\alpha} \rightarrow \frac{26\,351\,815\,728\,362\,500 - \sqrt{212\,915\,589\,402\,826\,661\,294\,025\,489\,275\,545}}{52\,703\,631\,456\,725\,000}, \right. \right.$$

$$P_{1,\alpha} \rightarrow \frac{23\,706\,303\,422\,472\,499 + \sqrt{212\,915\,589\,402\,826\,661\,294\,025\,489\,275\,545}}{49\,047\,621\,338\,677\,046},$$

$$P_{2,\alpha} \rightarrow \frac{19\,580\,987\,545\,285\,165 + \sqrt{212\,915\,589\,402\,826\,661\,294\,025\,489\,275\,545}}{45\,840\,001\,121\,224\,284},$$

$$P_{3,\alpha} \rightarrow \frac{28\,365\,918\,550\,451\,091 + \sqrt{212\,915\,589\,402\,826\,661\,294\,025\,489\,275\,545}}{59\,022\,948\,171\,315\,388},$$

$$P_{1,\beta} \rightarrow \frac{25\,341\,317\,916\,204\,547 + \sqrt{212\,915\,589\,402\,826\,661\,294\,025\,489\,275\,545}}{49\,047\,621\,338\,677\,046},$$

$$P_{2,\beta} \rightarrow \frac{26\,259\,013\,575\,939\,119 + \sqrt{212\,915\,589\,402\,826\,661\,294\,025\,489\,275\,545}}{45\,840\,001\,121\,224\,284},$$

$$P_{3,\beta} \rightarrow \frac{30\,657\,029\,620\,864\,297 + \sqrt{212\,915\,589\,402\,826\,661\,294\,025\,489\,275\,545}}{59\,022\,948\,171\,315\,388} \Big\},$$

$$\left\{ P_{\alpha} \rightarrow \frac{26\,351\,815\,728\,362\,500 + \sqrt{212\,915\,589\,402\,826\,661\,294\,025\,489\,275\,545}}{52\,703\,631\,456\,725\,000}, \right.$$

$$P_{1,\alpha} \rightarrow \frac{23\,706\,303\,422\,472\,499 - \sqrt{212\,915\,589\,402\,826\,661\,294\,025\,489\,275\,545}}{49\,047\,621\,338\,677\,046},$$

$$P_{2,\alpha} \rightarrow \frac{19\,580\,987\,545\,285\,165 - \sqrt{212\,915\,589\,402\,826\,661\,294\,025\,489\,275\,545}}{45\,840\,001\,121\,224\,284},$$

$$P_{3,\alpha} \rightarrow \frac{28\,365\,918\,550\,451\,091 - \sqrt{212\,915\,589\,402\,826\,661\,294\,025\,489\,275\,545}}{59\,022\,948\,171\,315\,388},$$

$$P_{1,\beta} \rightarrow \frac{25\,341\,317\,916\,204\,547 - \sqrt{212\,915\,589\,402\,826\,661\,294\,025\,489\,275\,545}}{49\,047\,621\,338\,677\,046},$$

$$P_{2,\beta} \rightarrow \frac{26\,259\,013\,575\,939\,119 - \sqrt{212\,915\,589\,402\,826\,661\,294\,025\,489\,275\,545}}{45\,840\,001\,121\,224\,284},$$

$$P_{3,\beta} \rightarrow \frac{30\,657\,029\,620\,864\,297 - \sqrt{212\,915\,589\,402\,826\,661\,294\,025\,489\,275\,545}}{59\,022\,948\,171\,315\,388} \Big\}$$

These are algebraic numbers since they are produced by exact algebraic polynomials. Mathematica is able to return them in this way because it uses Buchberger’s algorithm inside its Solve function. These exact numbers are hard to interpret. So lets get a floating point representation of them -

```
In[2228]:= sols // N
Out[2228]= { {Pα → 0.223138, P1,α → 0.780832, P2,α → 0.745476,
               P3,α → 0.727811, P1,β → 0.814167, P2,β → 0.891157, P3,β → 0.766628},
               {Pα → 0.776862, P1,α → 0.185833, P2,α → 0.108843, P3,α → 0.233372,
               P1,β → 0.219168, P2,β → 0.254524, P3,β → 0.272189} }
```

Let's pretty print them and compare them with what we used to simulate them.

```
In[2239]:= {First@sols // First /@# & // Rest,
             First@sols // N // Last /@# & // Rest,
             classifierAccuracies // Values /@# & // Transpose // Flatten} //
             Grid[#, Dividers → All] &
```

	P _{1,α}	P _{2,α}	P _{3,α}	P _{1,β}	P _{2,β}	P _{3,β}
Out[2239]=	0.780832	0.745476	0.727811	0.814167	0.891157	0.766628
	0.797697	0.738878	0.717426	0.819408	0.888438	0.757177

We'll leave you to ponder why the result is not closer to our distribution values.